Jackson Electrostatics - 2 09-16-17

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[8]:=
```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Purpose

This is the second of four notebooks dealing with electrostatic topics and calculations as treated in Jackson *Classical Electrodynamics.*

This notebook focuses on an electrostatic problem with axi-symmetry that is solved using an appropriate Green’s function. The treatment starts with a summary of Jackson’s solution of the Poisson equation for the Green’s function. Then the Green’s function is identified with the potential for a conducting sphere in the present of an external charge — obtained using the method of images in *Jackson Electrostatics - 1 09-16-17*. The Green’s function formalism is applied to the problem of a conducting sphere with hemispheres held at different potentials.

The general problem has no closed form solution. I treat special cases and approximate forms involving power-series expansions. I also solve the potential problem in term of radial functions and Legendre polynomials. I compare these approximation expressions against a solution obtained using numerical quadrature. Visualizations of equipotentials and the electric field are generated for both the interior and exterior solutions

1 Formal solution of Poisson equation

Roughly following Jackson 1.10 but with my own notation

A Green’s function $G(\mathbb{r}, \hat{\mathbb{r}})$ satisfies

$$\hat{\nabla}^2 G(\mathbb{r},\ \hat{\mathbb{r}}) = -4\,\pi\,\delta(\mathbb{r} - \hat{\mathbb{r}}) \tag{1}$$

within a volume V. The vector $\mathbb{r}$ is the position vector and $\hat{\mathbb{r}}$ denotes the location of a unit charge. The solution is

$$G(\mathbb{r},\ \hat{\mathbb{r}}) = \frac{1}{|\,\mathbb{r} - \hat{\mathbb{r}}\,|} + F(\mathbb{r},\ \hat{\mathbb{r}}) \tag{2}$$

where the first term is the particular solution associated with the point charge and the second term is the homogeneous solution

$$\hat{\nabla}^2 F(\mathbb{r},\ \hat{\mathbb{r}}) = 0 \tag{3}$$

within volume V.

Green's theorem can be used to write the formal solution of Poisson's equation

$$\Phi(\mathbb{r}) = \int_V d^3\,\hat{r}\,\rho(\hat{\mathbb{r}})\,G(\mathbb{r},\ \hat{\mathbb{r}}) + \frac{1}{4\,\pi}\oint dS\,G(\mathbb{r},\ \hat{\mathbb{r}})\,\frac{\partial\Phi}{\partial\hat{r}} - \frac{1}{4\,\pi}\oint dS\,\frac{\partial G(\mathbb{r},\ \hat{\mathbb{r}})}{\partial\hat{r}}\,\Phi(\hat{\mathbb{r}}) \tag{4}$$

There is a freedom in the form of G($\mathbb{r}$, $\hat{\mathbb{r}}$) in that the part F($\mathbb{r}$, $\hat{\mathbb{r}}$) can be chosen so that one or the other of the two surface integral vanishes. I am going to focus on Dirichlet boundary conditions in which F($\mathbb{r}$, $\hat{\mathbb{r}}$) is chosen to cause G($\mathbb{r}$, $\hat{\mathbb{r}}$) = 0 for $\hat{\mathbb{r}}$ on S. Thus

$$\Phi(\mathbb{r}) = \int_V d^3\,\hat{r}\,\rho(\hat{\mathbb{r}})\,G(\mathbb{r},\ \hat{\mathbb{r}}) - \frac{1}{4\,\pi}\oint dS\,\frac{\partial G(\mathbb{r},\ \hat{\mathbb{r}})}{\partial\hat{r}}\,\Phi(\hat{\mathbb{r}}) \tag{5}$$

2 Specific Green's function for conducting sphere

In particular, I will specialize the expression Φ($\mathbb{r}$) to the case of a conducting sphere

The geometry of the problem is

Out[12]=



The observation point is $\mathbb{r} = \{r, \theta, \phi\}$. The vector over which integrations are performed is $\hat{\mathbb{r}} = \{\hat{r}, \hat{\theta}, \hat{\phi}\}$

Some notational preliminaries.

In[3]:=

```
<< Notation`
```

In[4]:=

```
Symbolize[ 𝕣p ]; Symbolize[ qp ];
Symbolize[ rp ];
Symbolize[ 𝟙𝕣 ]; Symbolize[ 𝟙𝕣̂ ];
Symbolize[ 𝕣̂ ];
Symbolize[ θ̂ ];
Symbolize[ ϕ̂ ];
```

In notebook *Jackson Electrostatics - I 09-16-17*, the method of images was used to determine the potential associated with a grounded conducting sphere in the presence of an external charge

```
w2[1] = Φ[r] == -  a q_p / ( r_p Sqrt[r^2 + a^4/r_p^2 - 2 a^2 r Cos[θ]/r_p] ) + q_p / Sqrt[r^2 + r_p^2 - 2 r r_p Cos[θ]]
```

$$\Phi[\mathbb{r}] == -\frac{a\, q_p}{r_p \sqrt{r^2 + \frac{a^4}{r_p^2} - \frac{2\, a^2\, r\, \mathrm{Cos}[\theta]}{r_p}}} + \frac{q_p}{\sqrt{r^2 + r_p^2 - 2\, r\, r_p\, \mathrm{Cos}[\theta]}}$$

where r_p is the position of the point charge and θ is the angle between a radial line running through the observation point and the z-axis. The point charge and image charge were chosen, without loss of generality, to lie on the z-axis for purposes of simplifying that calculation.

To reexpress this as the Green's function for a conducting sphere, the following is required. The charge at $\hat{\mathbb{r}}$ is specified to be a unit charge (q_p = 1), the radial position of that charge is $r_p = \hat{r}$, where $\hat{\mathbb{r}} = \{\hat{r}, \hat{\theta}, \hat{\phi}\}$, the angle variable θ is generalized to correspond to ζ , the angle between $\mathbb{r}$ and $\hat{\mathbb{r}}$ as shown in the figure above.

```
w2[2] = w2[1] /. q_p → 1 /. r_p → r̂ /. θ → ζ
```

$$\Phi[\mathbb{r}] == -\frac{a}{\hat{r} \sqrt{r^2 + \frac{a^4}{\hat{r}^2} - \frac{2\, a^2\, r\, \mathrm{Cos}[\zeta]}{\hat{r}}}} + \frac{1}{\sqrt{r^2 + \hat{r}^2 - 2\, r\, \hat{r}\, \mathrm{Cos}[\zeta]}}$$

Specifically, $\zeta[\theta, \phi, \hat{\theta}, \hat{\phi}]$ is determined by taking the dot product of unit vectors in the $\mathbb{r}$ and $\hat{\mathbb{r}}$ directions.

```
w2[3] = Cos[ζ] == Dot[{Sin[θ] Cos[ϕ], Sin[θ] Sin[ϕ], Cos[θ]},
    {Sin[θ̂] Cos[ϕ̂], Sin[θ̂] Sin[ϕ̂], Cos[θ̂]}] // Simplify
```

$$\mathrm{Cos}[\zeta] == \mathrm{Cos}[\theta]\,\mathrm{Cos}[\hat{\theta}] + \mathrm{Cos}[\phi - \hat{\phi}]\,\mathrm{Sin}[\theta]\,\mathrm{Sin}[\hat{\theta}]$$

The second term in w2[2] is recognized as the first term in equation 2) — $\frac{1}{|\mathbb{r} - \hat{\mathbb{r}}|}$. The first term is $F(\mathbb{r}, \hat{\mathbb{r}})$, which must be chosen to satisfy the desired boundary conditions on the conducting sphere. But that is exactly what the image term in w2[2] does — it makes Φ = 0 on the surface. Thus we have $F(\mathbb{r}, \hat{\mathbb{r}}) = -\frac{a}{\sqrt{r^2 + \frac{a^4}{\hat{r}^2} - \frac{2\, a^2\, r\, \mathrm{Cos}[\zeta]}{\hat{r}}}\ \hat{r}}$ and

Thus the Green's function (for the Dirichlet boundary conditions Φ($\mathbb{r}$ on V) = 0) is

```
w2[4] = w2[2] /. Φ[r] → G[r, r̂]
```

$$G[\mathbb{r}, \hat{\mathbb{r}}] == -\frac{a}{\hat{r} \sqrt{r^2 + \frac{a^4}{\hat{r}^2} - \frac{2\, a^2\, r\, \mathrm{Cos}[\zeta]}{\hat{r}}}} + \frac{1}{\sqrt{r^2 + \hat{r}^2 - 2\, r\, \hat{r}\, \mathrm{Cos}[\zeta]}}$$

3) Application of Green’s function to a spherical conductor with hemispheres held at different potentials

I apply the Green’s function w2[4] to the problem of a “split” conducting sphere with its top hemisphere held at potential +V while its bottom hemisphere held at -V.

Sphere with top held at +V and bottom at –V



The Green’s function for this problem has Dirichlet boundary conditions so equation (4) becomes

$$\Phi(\mathbb{r}) = \int_V d^3\,\hat{r}\,\rho(\hat{\mathbb{r}})\,G(\mathbb{r},\,\hat{\mathbb{r}}) - \frac{1}{4\,\pi}\oint dS\,\frac{\partial G(\mathbb{r},\,\hat{\mathbb{r}})}{\partial\hat{r}}\,\Phi(\hat{\mathbb{r}}) \tag{6}$$

Further, there is no external charge density so

$$\Phi(\mathbb{r}) = -\frac{1}{4\,\pi}\oint dS\,\frac{\partial G(\mathbb{r},\,\hat{\mathbb{r}})}{\partial\hat{r}}\,\Phi(\hat{\mathbb{r}}) \tag{7}$$

From the previous section the Green’s function is

```
w3[1] = w2[4]
```

$$G[\mathbb{r},\,\hat{\mathbb{r}}] == -\frac{a}{\hat{r}\sqrt{r^2 + \frac{a^4}{\hat{r}^2} - \frac{2\,a^2\,r\,\mathrm{Cos}[\zeta]}{\hat{r}}}} + \frac{1}{\sqrt{r^2 + \hat{r}^2 - 2\,r\,\hat{r}\,\mathrm{Cos}[\zeta]}}$$

I first determine the potential outside the sphere. The derivative $\frac{\partial G}{\partial \hat{n}}$ should be directed normal to the surface of the sphere and away from the region of interest, i.e., the region outside the sphere. Since the unit normal for the surface element is outward ($+\hat{\mathbb{r}}$) the direction for $\frac{\partial G}{\partial \hat{r}}$ should inward ($-\hat{\mathbb{r}}$)

```
w3[2] = dG[𝕣, 𝕣̂] == D[w3[1][[2]], r̂] /. r̂ → a // Simplify
```

$$\mathrm{dG}[\mathbb{r}, \hat{\mathbb{r}}] == \frac{-a^2 + r^2}{a\left(a^2 + r^2 - 2\,a\,r\,\mathrm{Cos}[\zeta]\right)^{3/2}}$$

The integral (7) to be performed is

```
w3[3] = -1/(4 π) Inactive[Integrate][Φ[a, θ̂, ϕ̂] dG[𝕣, 𝕣̂] (-1) (a^2 Sin[θ̂]),
    {θ̂, 0, π}, {ϕ̂, 0, 2 π}, Assumptions → {a > 0, r > a}]
```

$$-\frac{1}{4\pi}\mathrm{Integrate}\left[-a^2\,\mathrm{dG}[\mathbb{r}, \hat{\mathbb{r}}]\,\mathrm{Sin}[\hat{\theta}]\,\Phi[a, \hat{\theta}, \hat{\phi}],\ \{\hat{\theta}, 0, \pi\}, \{\hat{\phi}, 0, 2\pi\}, \mathrm{Assumptions} \to \{a > 0, r > a\}\right]$$

where the (-1) is because dG $(-\mathbb{1}_{\mathbb{r}}) \cdot$ dA $(\mathbb{1}_{\mathbb{r}})$ = (-1) dG $\left(a^2 \sin(\hat{\theta})\, d\hat{\theta}\, d\hat{\phi}\right)$. Also, I anticipate that the assumptions on a and r should be imposed to avoid problems associated with selecting an inappropriate branch. I make the integral “Inactive” to suppress immediate attempts at evaluation.

In detail,

```
w3[4] = w3[3] /. (w3[2] // ER)
```

$$-\frac{1}{4\pi}\mathrm{Integrate}\left[-\left(\left(a\left(-a^2 + r^2\right)\mathrm{Sin}[\hat{\theta}]\,\Phi[a, \hat{\theta}, \hat{\phi}]\right) \Big/ \left(a^2 + r^2 - 2\,a\,r\,\mathrm{Cos}[\zeta]\right)^{3/2}\right),\ \{\hat{\theta}, 0, \pi\}, \{\hat{\phi}, 0, 2\pi\}, \mathrm{Assumptions} \to \{a > 0, r > a\}\right]$$

For the problem at hand, Φ[a, $\hat{\theta}$, $\hat{\phi}$] = {V for 0 < $\hat{\theta}$ < π/2, -V for π/2 < $\hat{\theta}$ < π}

```
w3[5] = (w3[4] /. Φ[a, θ̂, ϕ̂] → V /. {θ̂, 0, π} → {θ̂, 0, π/2}) +
   (w3[4] /. Φ[a, θ̂, ϕ̂] → -V /. {θ̂, 0, π} → {θ̂, π/2, π})
```

$$-\frac{1}{4\pi}\mathrm{Integrate}\left[-\left(\left(a\left(-a^2 + r^2\right)V\,\mathrm{Sin}[\hat{\theta}]\right) \Big/ \left(a^2 + r^2 - 2\,a\,r\,\mathrm{Cos}[\zeta]\right)^{3/2}\right),\ \left\{\hat{\theta}, 0, \frac{\pi}{2}\right\}, \{\hat{\phi}, 0, 2\pi\}, \mathrm{Assumptions} \to \{a > 0, r > a\}\right] - \frac{1}{4\pi}\mathrm{Integrate}\left[\frac{a\left(-a^2 + r^2\right)V\,\mathrm{Sin}[\hat{\theta}]}{\left(a^2 + r^2 - 2\,a\,r\,\mathrm{Cos}[\zeta]\right)^{3/2}},\ \left\{\hat{\theta}, \frac{\pi}{2}, \pi\right\}, \{\hat{\phi}, 0, 2\pi\}, \mathrm{Assumptions} \to \{a > 0, r > a\}\right]$$

where it should be remembered ζ has a complicated form (w2[3] above).

As it stands, these integrals cannot be evaluated in closed form. However, special cases are tractable. I consider some of these below and also check the results against a numerical evaluation of the integral w3[5].

3.1 Special case: solution along the +z-axis

A tractable special case is when the observation point is along the z-axis, i.e., $\theta = 0$ and r = z. Consider

```
w2[3]
```

$$\text{Cos}[\zeta] == \text{Cos}[\theta]\,\text{Cos}[\hat{\theta}] + \text{Cos}[\phi - \hat{\phi}]\,\text{Sin}[\theta]\,\text{Sin}[\hat{\theta}]$$

```
w31[1] = w2[3] /. θ → 0
```

$$\text{Cos}[\zeta] == \text{Cos}[\hat{\theta}]$$

```
w31[2] = w3[5] /. (w31[1] // ER) /. r → z
```

$$-\frac{1}{4\pi}\text{Integrate}\left[-\left(\left(a\,V\left(-a^2+z^2\right)\text{Sin}[\hat{\theta}]\right)\Big/\left(a^2+z^2-2\,a\,z\,\text{Cos}[\hat{\theta}]\right)^{3/2}\right),\left\{\hat{\theta},0,\frac{\pi}{2}\right\},\left\{\hat{\phi},0,2\pi\right\},\text{Assumptions}\to\{a>0,z>a\}\right]-\frac{1}{4\pi}\text{Integrate}\left[\frac{a\,V\left(-a^2+z^2\right)\text{Sin}[\hat{\theta}]}{\left(a^2+z^2-2\,a\,z\,\text{Cos}[\hat{\theta}]\right)^{3/2}},\left\{\hat{\theta},\frac{\pi}{2},\pi\right\},\left\{\hat{\phi},0,2\pi\right\},\text{Assumptions}\to\{a>0,z>a\}\right]$$

These integrals can be evaluated.

```
w31[3] = Φtop == Activate[w31[2][[1]]]
```

$$\Phi\text{top} == \frac{V\,(a+z)\left(a-z+\sqrt{a^2+z^2}\right)}{2\,z\,\sqrt{a^2+z^2}}$$

```
w31[4] = Φbottom = Activate[w31[2][[2]]]
```

$$-\frac{V\,(a-z)\left(-a-z+\sqrt{a^2+z^2}\right)}{2\,z\,\sqrt{a^2+z^2}}$$

These expressions agree with Jackson 2.28

Since the apparatus is in place, I determine the interior solution by changing the sign of $\frac{\partial G}{\partial \hat{r}}$ and assuming r < a. These changes can be effected with a rule

```
w31[5] = w31[2] /.
  Inactive[Integrate][integrand_, lim1_, lim2_, Assumptions → {a > 0, z > a}] →
   Inactive[Integrate][-integrand, lim1, lim2, Assumptions → {a > 0, z < a}]
```

$$-\frac{1}{4\pi}\text{Integrate}\left[-\left(\left(a V\left(-a^2+z^2\right) \text{Sin}[\hat{\theta}]\right)\Big/\left(a^2+z^2-2 a z \text{Cos}[\hat{\theta}]\right)^{3/2}\right),\left\{\hat{\theta},\frac{\pi}{2},\pi\right\},\left\{\hat{\phi},0,2\pi\right\},\text{Assumptions}\to\{a>0,z<a\}\right]-\frac{1}{4\pi}\text{Integrate}\left[\frac{a V\left(-a^2+z^2\right)\text{Sin}[\hat{\theta}]}{\left(a^2+z^2-2 a z \text{Cos}[\hat{\theta}]\right)^{3/2}},\left\{\hat{\theta},0,\frac{\pi}{2}\right\},\left\{\hat{\phi},0,2\pi\right\},\text{Assumptions}\to\{a>0,z<a\}\right]$$

As before

```
w31[6] = Activate /@ w31[5]
```

$$\frac{V(a+z)\left(-a+z+\sqrt{a^2+z^2}\right)}{2 z\sqrt{a^2+z^2}}-\left(V(a-z)\left(a+z-\sqrt{a^2+z^2}\,\text{Sign}[a+z]\right)\right)\Big/\left(2 z\sqrt{a^2+z^2}\right)$$

I define functions for these results

```
Clear[ΦSphere, ΦTopHemisphere, ΦBottomHemisphere];
ΦTopHemisphere[z_, a_, V_] :=
  ((V (a + z) (-a + z + Sqrt[a^2 + z^2]))/(2 z Sqrt[a^2 + z^2])) HeavisideTheta[a - z] +
   ((V (a + z) (a - z + Sqrt[a^2 + z^2]))/(2 z Sqrt[a^2 + z^2])) HeavisideTheta[z - a];
ΦBottomHemisphere[z_, a_, V_] :=
  -((V (a - z) (a + z - Sqrt[a^2 + z^2] Sign[a + z]))/(2 z Sqrt[a^2 + z^2])) HeavisideTheta[a - z] -
   ((V (a - z) (-a - z + Sqrt[a^2 + z^2]))/(2 z Sqrt[a^2 + z^2])) HeavisideTheta[z - a];
ΦSphere[z_, a_, V_] := ΦTopHemisphere[z, a, V] + ΦBottomHemisphere[z, a, V]
```

where I have used Heaviside functions to allow the interior and exterior expressions to be handled with one function.

For representative parameters

```
Module[{a = 1, V = 1, lab},
 lab = Stl[StringForm["Potential along +z-axis"]];
 Plot[{ΦSphere[z, a, V], ΦTopHemisphere[z, a, V], ΦBottomHemisphere[z, a, V]},
  {z, 0, 3 a}, AxesLabel → {Stl["z"], Stl["Potential"]},
  PlotStyle → {Black, Blue, Red}, PlotLegends →
   {"complete sphere", "top hemisphere", "bottom hemisphere"}, PlotLabel → lab]]
```



3.2 Special case: power series expansion along the +z-axis

Recall the general integral

```
w3[5]
```

$$-\frac{1}{4\pi}\text{Integrate}\left[-\left(\left(a\left(-a^2+r^2\right)V\,\text{Sin}[\hat{\theta}]\right)\Big/\left(a^2+r^2-2\,a\,r\,\text{Cos}[\zeta]\right)^{3/2}\right),\right.$$
$$\left.\left\{\hat{\theta},0,\frac{\pi}{2}\right\},\left\{\hat{\phi},0,2\pi\right\},\text{Assumptions}\to\{a>0,r>a\}\right]-\frac{1}{4\pi}$$
$$\text{Integrate}\left[\frac{a\left(-a^2+r^2\right)V\,\text{Sin}[\hat{\theta}]}{\left(a^2+r^2-2\,a\,r\,\text{Cos}[\zeta]\right)^{3/2}},\left\{\hat{\theta},\frac{\pi}{2},\pi\right\},\left\{\hat{\phi},0,2\pi\right\},\text{Assumptions}\to\{a>0,r>a\}\right]$$

I develop a solution appropriate for z > a. Consider the denominator

```
w32[1] = (a^2 + r^2 - 2 a r Cos[ζ])^(-3/2)
```

$$\frac{1}{\left(a^2+r^2-2\,a\,r\,\text{Cos}[\zeta]\right)^{3/2}}$$

and define

```
def[α] = α == 2 a r/(a^2 + r^2)
```

$$\alpha == \frac{2\,a\,r}{a^2 + r^2}$$

Rather than making some tricky substitutions, I will just write

```
w32[2] = 1/((a^2 + r^2)^(3/2) (1 - α Cos[ζ])^(3/2))
```

$$\frac{1}{\left(a^2+r^2\right)^{3/2}\,(1-\alpha\,\mathrm{Cos}[\zeta])^{3/2}}$$

Expanding

```
w32[3] = Series[w32[2], {α, 0, 3}]
```

$$\frac{1}{\left(a^2+r^2\right)^{3/2}}+\frac{3\,\mathrm{Cos}[\zeta]\,\alpha}{2\left(a^2+r^2\right)^{3/2}}+\frac{15\,\mathrm{Cos}[\zeta]^2\,\alpha^2}{8\left(a^2+r^2\right)^{3/2}}+\frac{35\,\mathrm{Cos}[\zeta]^3\,\alpha^3}{16\left(a^2+r^2\right)^{3/2}}+O[\alpha]^4$$

Then, the integrals become

```
w32[4] = w3[5] /. (a^2 + r^2 - 2 a r Cos[ζ])^(-3/2) → Normal@w32[3]
```

$$-\frac{1}{4\pi}\,\mathrm{Integrate}\Big[-a\left(-a^2+r^2\right)V\left(\frac{1}{\left(a^2+r^2\right)^{3/2}}+\frac{3\,\alpha\,\mathrm{Cos}[\zeta]}{2\left(a^2+r^2\right)^{3/2}}+\frac{15\,\alpha^2\,\mathrm{Cos}[\zeta]^2}{8\left(a^2+r^2\right)^{3/2}}+\frac{35\,\alpha^3\,\mathrm{Cos}[\zeta]^3}{16\left(a^2+r^2\right)^{3/2}}\right)$$
$$\mathrm{Sin}[\hat{\theta}],\ \left\{\hat{\theta},0,\frac{\pi}{2}\right\},\ \left\{\hat{\phi},0,2\pi\right\},\ \mathrm{Assumptions}\to\{a>0,\ r>a\}\Big]-$$
$$\frac{1}{4\pi}\,\mathrm{Integrate}\Big[a\left(-a^2+r^2\right)V\left(\frac{1}{\left(a^2+r^2\right)^{3/2}}+\frac{3\,\alpha\,\mathrm{Cos}[\zeta]}{2\left(a^2+r^2\right)^{3/2}}+\frac{15\,\alpha^2\,\mathrm{Cos}[\zeta]^2}{8\left(a^2+r^2\right)^{3/2}}+\frac{35\,\alpha^3\,\mathrm{Cos}[\zeta]^3}{16\left(a^2+r^2\right)^{3/2}}\right)$$
$$\mathrm{Sin}[\hat{\theta}],\ \left\{\hat{\theta},\frac{\pi}{2},\pi\right\},\ \left\{\hat{\phi},0,2\pi\right\},\ \mathrm{Assumptions}\to\{a>0,\ r>a\}\Big]$$

To simplify the presentation, I just consider the first integral valid for the top hemisphere

```
w32[5] = w32[4][[1]]
```

$$-\frac{1}{4\pi}\,\mathrm{Integrate}\Big[-a\left(-a^2+r^2\right)V\left(\frac{1}{\left(a^2+r^2\right)^{3/2}}+\frac{3\,\alpha\,\mathrm{Cos}[\zeta]}{2\left(a^2+r^2\right)^{3/2}}+\frac{15\,\alpha^2\,\mathrm{Cos}[\zeta]^2}{8\left(a^2+r^2\right)^{3/2}}+\frac{35\,\alpha^3\,\mathrm{Cos}[\zeta]^3}{16\left(a^2+r^2\right)^{3/2}}\right)$$
$$\mathrm{Sin}[\hat{\theta}],\ \left\{\hat{\theta},0,\frac{\pi}{2}\right\},\ \left\{\hat{\phi},0,2\pi\right\},\ \mathrm{Assumptions}\to\{a>0,\ r>a\}\Big]$$

Introduce the explicit expression for ζ

```
w32[6] = w32[5] /. (w2[3] // ER)
```

$$-\frac{1}{4\pi}\text{Integrate}\Big[-a\left(-a^2+r^2\right)V\,\text{Sin}[\hat{\theta}]\left(\frac{1}{\left(a^2+r^2\right)^{3/2}}+\left(3\alpha\left(\text{Cos}[\theta]\,\text{Cos}[\hat{\theta}]+\text{Cos}[\phi-\hat{\phi}]\,\text{Sin}[\theta]\,\text{Sin}[\hat{\theta}]\right)\right)\Big/\left(2\left(a^2+r^2\right)^{3/2}\right)+\left(15\alpha^2\left(\text{Cos}[\theta]\,\text{Cos}[\hat{\theta}]+\text{Cos}[\phi-\hat{\phi}]\,\text{Sin}[\theta]\,\text{Sin}[\hat{\theta}]\right)^2\right)\Big/\left(8\left(a^2+r^2\right)^{3/2}\right)+\left(35\alpha^3\left(\text{Cos}[\theta]\,\text{Cos}[\hat{\theta}]+\text{Cos}[\phi-\hat{\phi}]\,\text{Sin}[\theta]\,\text{Sin}[\hat{\theta}]\right)^3\right)\Big/\left(16\left(a^2+r^2\right)^{3/2}\right)\right),\left\{\hat{\theta},0,\frac{\pi}{2}\right\},\left\{\hat{\phi},0,2\pi\right\},\text{Assumptions}\to\{a>0,r>a\}\Big]$$

With some additional symbolic manipulation it would be possible to follow the treatment in Jackson and isolate individual integrals, invoke symmetries, etc. However, I just take advantage of Mathematica and evaluate the integral. It it takes a while.

```
w32[7] = Timing[Activate[ w32[6]]]
```

$$\left\{73.8125,-\left(\left(a\left(a^2-r^2\right)V\left(64\left(8+5\alpha^2\right)+3\alpha\left(128+105\alpha^2\right)\text{Cos}[\theta]-35\alpha^3\,\text{Cos}[3\theta]\right)\right)\Big/\left(1024\left(a^2+r^2\right)^{3/2}\right)\right)\right\}$$

In a similar manner, I calculate the contribution from the bottom hemisphere

```
w32[8] = w32[4][[2]] /. (w2[3] // ER) // Activate
```

$$\left(a\left(a^2-r^2\right)V\left(64\left(8+5\alpha^2\right)-3\alpha\left(128+105\alpha^2\right)\text{Cos}[\theta]+35\alpha^3\,\text{Cos}[3\theta]\right)\right)\Big/\left(1024\left(a^2+r^2\right)^{3/2}\right)$$

Combining the contributions from the two hemispheres

```
w32[9] = w32[7][[2]] + w32[8] // Simplify
```

$$\left(a\left(a^2-r^2\right)V\alpha\,\text{Cos}[\theta]\left(-192-175\alpha^2+35\alpha^2\,\text{Cos}[2\theta]\right)\right)\Big/\left(256\left(a^2+r^2\right)^{3/2}\right)$$

or, isolating the α dependence,

```
w32[10] = Factor /@ Collect[Expand[w32[9]], α]
```

$$-\left(\left(3a\,(a-r)\,(a+r)\,V\alpha\,\text{Cos}[\theta]\right)\Big/\left(4\left(a^2+r^2\right)^{3/2}\right)\right)+\left(35a\,(a-r)\,(a+r)\,V\alpha^3\,\text{Cos}[\theta]\left(-5+\text{Cos}[2\theta]\right)\right)\Big/\left(256\left(a^2+r^2\right)^{3/2}\right)$$

I define a function for this expression

```
Clear[ΦApprox];
ΦApprox[r_, θ_, V_, a_] :=
 Module[{α = (2 a r)/(a^2 + r^2)},
  -((3 a (a - r) (a + r) V α Cos[θ]) / (4 (a^2 + r^2)^(3/2))) +
   (35 a (a - r) (a + r) V α^3 Cos[θ] (-5 + Cos[2 θ])) / (256 (a^2 + r^2)^(3/2))]
```

Check this result against the exact result available for $\theta = 0$

```
Module[{a = 1, V = 1, θ = 0},
 Plot[{ΦSphere[r, a, V], ΦApprox[r, θ, V, a], (2 a r)/(a^2 + r^2)},
  {r, 1, 5}, PlotRange → All, AxesLabel → {Stl["z"], Stl["Potential"]},
  PlotStyle → {Blue, Directive[Blue, Dashed], LightGray},
  PlotLegends → {"exact for θ = 0", "2 term approximation", "value of α"},
  PlotLabel → Stl["Comparison of Φapprox with exact result for θ = 0"]]]
```



The two-term approximation becomes accurate for z ≃ 3, even though $\alpha \simeq 0.6$. With some effort, rules could be developed for faster evaluation of the integration above and arbitrarily higher order terms in α could be calculated.

3.3 Numerical integration

I prepare expressions for the numerical calculation

```
w33[1] = w3[5] /. (w2[3] // ER) /. {θ̂ → θh, ϕ̂ → ϕh}
```

$$-\frac{1}{4\pi}\text{Integrate}\Big[-\Big(\big(a\,(-a^2+r^2)\,V\,\text{Sin}[\theta h]\big)\Big/\big(a^2+r^2-2\,a\,r\,(\text{Cos}[\theta]\,\text{Cos}[\theta h]+\text{Cos}[\phi-\phi h]\,\text{Sin}[\theta]\,\text{Sin}[\theta h])\big)^{3/2}\Big),\ \Big\{\theta h,0,\frac{\pi}{2}\Big\},\ \{\phi h,0,2\pi\},\ \text{Assumptions}\to\{a>0,r>a\}\Big]-\frac{1}{4\pi}\text{Integrate}\Big[\big(a\,(-a^2+r^2)\,V\,\text{Sin}[\theta h]\big)\Big/\big(a^2+r^2-2\,a\,r\,(\text{Cos}[\theta]\,\text{Cos}[\theta h]+\text{Cos}[\phi-\phi h]\,\text{Sin}[\theta]\,\text{Sin}[\theta h])\big)^{3/2},\ \Big\{\theta h,\frac{\pi}{2},\pi\Big\},\ \{\phi h,0,2\pi\},\ \text{Assumptions}\to\{a>0,r>a\}\Big]$$

```
w33[2] = w33[1] /. Inactive[Integrate][a_, b_, c_, d_] → NIntegrate[a, b, c]
```

$$-\frac{1}{4\pi}\text{NIntegrate}\Big[-\Big(\big(a\,(-a^2+r^2)\,V\,\text{Sin}[\theta h]\big)\Big/\big(a^2+r^2-2\,a\,r\,(\text{Cos}[\theta]\,\text{Cos}[\theta h]+\text{Cos}[\phi-\phi h]\,\text{Sin}[\theta]\,\text{Sin}[\theta h])\big)^{3/2}\Big),\ \Big\{\theta h,0,\frac{\pi}{2}\Big\},\ \{\phi h,0,2\pi\}\Big]-\frac{1}{4\pi}\text{NIntegrate}\Big[\big(a\,(-a^2+r^2)\,V\,\text{Sin}[\theta h]\big)\Big/\big(a^2+r^2-2\,a\,r\,(\text{Cos}[\theta]\,\text{Cos}[\theta h]+\text{Cos}[\phi-\phi h]\,\text{Sin}[\theta]\,\text{Sin}[\theta h])\big)^{3/2},\ \Big\{\theta h,\frac{\pi}{2},\pi\Big\},\ \{\phi h,0,2\pi\}\Big]$$

```
Clear[ΦNumerical];
ΦNumerical[r_, θ_, ϕ_, {a_, V_}] :=
 -1/(4 π) NIntegrate[-((a (-a^2 + r^2) V Sin[θh]) /
        (a^2 + r^2 - 2 a r (Cos[θ] Cos[θh] + Cos[ϕ - ϕh] Sin[θ] Sin[θh]))^(3/2)),
     {θh, 0, π/2}, {ϕh, 0, 2 π}] - 1/(4 π) NIntegrate[(a (-a^2 + r^2) V Sin[θh]) /
    (a^2 + r^2 - 2 a r (Cos[θ] Cos[θh] + Cos[ϕ - ϕh] Sin[θ] Sin[θh]))^(3/2),
   {θh, π/2, π}, {ϕh, 0, 2 π}]
```

Check that the numerical calculation reproduces the exact solution in the limit $\theta \to 0$

```
Module[{a = 1, V = 1, θ = 0, ϕ = 0, r = 3, vals, lab},
 vals = Table[{r, ΦNumerical[r, θ, ϕ, {a, V}]}, {r, a + 0.0001, 5 a, a / 4}];
 lab = Stl@StringForm["Check of numerical result against on-axis exact result"];
 Plot[ΦSphere[r, a, V], {r, a + 0.0001, 5}, Epilog → {OC[#, Black] & /@ vals},
  AxesLabel → {Stl["z"], Stl["Φ"]}, PlotLabel → lab]]
```



where the line is the exact closed form result and the circles are from numerical quadrature.

I also check the accuracy of the θ dependence of the two-term approximation

```
Module[{a = 1, V = 1, θ = 0, ϕ = 0, r = 3, vals, lab},
 vals = Table[{θ, ΦNumerical[r, θ, ϕ, {a, V}]}, {θ, 0, π, π / 8}];
 lab = Stl["Comparing two-term approximation against numerical solution"];
 Plot[ ΦApprox[r, θ, V, a], {θ, 0, π}, Epilog → {OC[#, Black] & /@ vals},
  AxesLabel → {Stl["θ"], Stl["Φ"]}, PlotLabel → lab]]
```



3.4 Solution in terms of orthogonal functions

Laplace' s partial differential equation can be solved using separation of variables. The derivation is standard and discussed in numerous texts. I will just start with the series solution and apply it to the hemisphere problem.

The hemisphere problem has azimuthal symmetry and the non-ignorable variables are r and θ. The general solution is

```
w34[1] = Φ[r, θ] == Sm[(Aℓ r^ℓ + Bℓ r^-(ℓ+1)) LegendreP[ℓ, Cos[θ]], {ℓ, 0, ∞}]
```

$$\Phi[r, \theta] == \text{Sm}\left[\text{LegendreP}[\ell, \text{Cos}[\theta]]\left(r^{\ell} A_{\ell} + r^{-1-\ell} B_{\ell}\right), \{\ell, 0, \infty\}\right]$$

where I have chosen my own function $\mathcal{S}$ instead of Mathematica's Sum so as to have more control over when the expression evaluates.

In this particular problem the boundary condition is θ dependent and will be applied at the surface of the conductor r = a. I rewrite this equation in terms of a general function f[r, ℓ] so that I can solve both the interior problem (r < a) and exterior problem r > a using the same operations.

```
w34[2] = Φ[r, θ] == 𝒮[ℓ][Aℓ f[r, ℓ] LegendreP[ℓ, Cos[θ]], {ℓ, 0, ∞}]
```

$$\Phi[r, \theta] == \mathcal{S}[\ell][f[r, \ell]\, \text{LegendreP}[\ell, \text{Cos}[\theta]]\, A_{\ell}, \{\ell, 0, \infty\}]$$

The boundary conditio at the surface of the sphere is that Φ[a, θ] = V for 0 < θ < π/2 and -V for π/2 < 0 < π

```
w34[3] = w34[2] /. r → a
```

$$\Phi[a, \theta] == \mathcal{S}[\ell][f[a, \ell]\, \text{LegendreP}[\ell, \text{Cos}[\theta]]\, A_{\ell}, \{\ell, 0, \infty\}]$$

Operate on this expression with

```
w34[4] = MapEqn[ℐ[θ][Sin[θ] LegendreP[k, Cos[θ]] #, {θ, 0, π}] &, w34[3]]
```

$$\mathcal{I}[\theta][\text{LegendreP}[k, \text{Cos}[\theta]]\, \text{Sin}[\theta]\, \Phi[a, \theta], \{\theta, 0, \pi\}] == \mathcal{I}[\theta][\text{LegendreP}[k, \text{Cos}[\theta]]\, \text{Sin}[\theta]\, \mathcal{S}[\ell][f[a, \ell]\, \text{LegendreP}[\ell, \text{Cos}[\theta]]\, A_{\ell}, \{\ell, 0, \infty\}], \{\theta, 0, \pi\}]$$

where $\mathcal{I}[\theta]$[f] is my operator for Integrate[f, {θ, 0, π}]. Focus on the rhs and interchange the integral and sum

```
rhs[1] = w34[4][[2]] /.
  ℐ[θ][a_ 𝒮[ℓ][b_, {ℓ, 0, ∞}], {θ, 0, π}] → 𝒮[ℓ][ℐ[θ][a b, {θ, 0, π}], {ℓ, 0, ∞}]
```

$$\mathcal{S}[\ell][\mathcal{I}[\theta][f[a, \ell]\, \text{LegendreP}[k, \text{Cos}[\theta]]\, \text{LegendreP}[\ell, \text{Cos}[\theta]]\, \text{Sin}[\theta]\, A_{\ell}, \{\theta, 0, \pi\}], \{\ell, 0, \infty\}]$$

Invoke the orthogonality of the Legendre polynomials

```
rhs[2] =
 rhs[1] /. ℐ[θ][a_ LegendreP[k, Cos[θ]] LegendreP[ℓ, Cos[θ]] Sin[θ], {θ, 0, π}] /;
    FreeQ[a, θ] → a 2/(2 ℓ + 1) KroneckerDelta[ℓ, k]
```

$\mathcal{S}[\ell]\left[\frac{1}{1+2\ell}2\,f[a,\ell]\ \text{KroneckerDelta}[k,\ell]\ A_\ell,\ \{\ell,0,\infty\}\right]$

Thus

```
w34[5] = w34[4][[1]] == rhs[2]
```

$\mathcal{I}[\theta]\left[\text{LegendreP}[k,\text{Cos}[\theta]]\ \text{Sin}[\theta]\ \Phi[a,\theta],\ \{\theta,0,\pi\}\right] == \mathcal{S}[\ell]\left[\frac{1}{1+2\ell}2\,f[a,\ell]\ \text{KroneckerDelta}[k,\ell]\ A_\ell,\ \{\ell,0,\infty\}\right]$

Impose the explicit boundary condition

```
w34[6] = w34[5] /. Φ[a, θ] → If[0 < θ < π/2, V, -V]
```

$\mathcal{I}[\theta]\left[\text{If}\left[0<\theta<\frac{\pi}{2},V,-V\right]\ \text{LegendreP}[k,\text{Cos}[\theta]]\ \text{Sin}[\theta],\ \{\theta,0,\pi\}\right] == \mathcal{S}[\ell]\left[\frac{1}{1+2\ell}2\,f[a,\ell]\ \text{KroneckerDelta}[k,\ell]\ A_\ell,\ \{\ell,0,\infty\}\right]$

I will manually demonstrate the solution to order 5, but then write a function to perform the calculation for an arbitrary number of expansion terms.

```
w34[7] = Table[w34[6] /. ℐ[θ] → Integrate /. 𝒮[ℓ] → Sum, {k, 0, 5}]
```

$\left\{0 == 2\,f[a,0]\,A_0,\ V == \frac{2}{3}f[a,1]\,A_1,\ 0 == \frac{2}{5}f[a,2]\,A_2,\ -\frac{V}{4} == \frac{2}{7}f[a,3]\,A_3,\ 0 == \frac{2}{9}f[a,4]\,A_4,\ \frac{V}{8} == \frac{2}{11}f[a,5]\,A_5\right\}$

```
w34[8] = Solve[w34[7], Table[A_i, {i, 0, 5}]][[1]]
```

$\left\{A_0\to 0,\ A_1\to\frac{3V}{2\,f[a,1]},\ A_2\to 0,\ A_3\to-\frac{7V}{8\,f[a,3]},\ A_4\to 0,\ A_5\to\frac{11V}{16\,f[a,5]}\right\}$

```
w34[9] = 𝒮[ℓ][f[r, ℓ] Inactive[LegendreP][ℓ, Cos[θ]] A_ℓ,
    {ℓ, 0, Length[w34[8]] - 1}] /. 𝒮[ℓ] → Sum /. w34[8]
```

$(3\,V\,f[r,1]\ \text{LegendreP}[1,\text{Cos}[\theta]])/(2\,f[a,1]) - (7\,V\,f[r,3]\ \text{LegendreP}[3,\text{Cos}[\theta]])/(8\,f[a,3]) + (11\,V\,f[r,5]\ \text{LegendreP}[5,\text{Cos}[\theta]])/(16\,f[a,5])$

Then, the interior and exterior solutions are

```
w34[10] = {w34[9] /. f[r_, ℓ_] → r^ℓ, w34[9] /. f[r_, ℓ_] → r^-(ℓ+1)}
```

$$\left\{\frac{3\,r\,V\,\text{LegendreP}[1,\text{Cos}[\theta]]}{2\,a}-\frac{1}{8\,a^3}7\,r^3\,V\,\text{LegendreP}[3,\text{Cos}[\theta]]+\frac{1}{16\,a^5}11\,r^5\,V\,\text{LegendreP}[5,\text{Cos}[\theta]],\ \frac{1}{2\,r^2}3\,a^2\,V\,\text{LegendreP}[1,\text{Cos}[\theta]]-\frac{1}{8\,r^4}7\,a^4\,V\,\text{LegendreP}[3,\text{Cos}[\theta]]+\frac{1}{16\,r^6}11\,a^6\,V\,\text{LegendreP}[5,\text{Cos}[\theta]]\right\}$$

The first expression is Jackson 3.37.

I write a function to perform this calculation to n terms for arbitrary potential Φ(a, θ)

```
Clear[GenerateSolution];
GenerateSolution[Φbc_, nTerms_] :=
 Module[{eqns, coeffs},
  eqns = Table[ℐ[θ][Φbc LegendreP[k, Cos[θ]] Sin[θ], {θ, 0, π}] ==
       𝒮[ℓ][(2 f[a, ℓ] KroneckerDelta[k, ℓ] Aℓ) / (1 + 2 ℓ), {ℓ, 0, ∞}] /.
      ℐ[θ] → Integrate /. 𝒮[ℓ] → Sum, {k, 0, nTerms}];
  coeffs = Solve[eqns, Table[Ai, {i, 0, nTerms}]][[1]];
  Sum[f[r, ℓ] Inactive[LegendreP][ℓ, Cos[θ]] Aℓ, {ℓ, 0, nTerms}] /. coeffs ]
```

Then, for example, to order 7

```
w34[11] = GenerateSolution[If[0 < θ < π/2, V, -V], 7]
```

$$(3\,V\,f[r,1]\,\text{LegendreP}[1,\text{Cos}[\theta]])/(2\,f[a,1])-(7\,V\,f[r,3]\,\text{LegendreP}[3,\text{Cos}[\theta]])/(8\,f[a,3])+(11\,V\,f[r,5]\,\text{LegendreP}[5,\text{Cos}[\theta]])/(16\,f[a,5])-(75\,V\,f[r,7]\,\text{LegendreP}[7,\text{Cos}[\theta]])/(128\,f[a,7])$$

I create an Association data structure to store the information associated with this expansion. This allows the expansion terms to be precalculated, transformed and stored for more efficient access in subsequent calculations.

```
Clear[AExpansion];
AExpansion = Module[{ℓMax = 21, terms, termsInner,
    termsOuter, termsInnerTransformed, termsOuterTransformed},
   terms = GenerateSolution[If[0 < θ < π/2, V, -V], ℓMax] /. Plus → List;
   termsInner = terms /. f[r_, ℓ_] → r^ℓ;
   termsOuter = terms /. f[r_, ℓ_] → r^-(ℓ+1);
   termsInnerTransformed =
    TransformedField["Spherical" → "Cartesian", #, {r, θ, φ} → {x, y, z}] & /@
     Activate[termsInner];
   termsOuterTransformed = TransformedField["Spherical" → "Cartesian",
       #, {r, θ, φ} → {x, y, z}] & /@ Activate[termsOuter];
   Association["terms" → terms, "termsInnerrθϕ" → termsInner,
    "termsInnerxyz" → termsInnerTransformed, "termsOuterrθϕ" → termsOuter,
    "termsOuterxyz" → termsOuterTransformed]];
```

Then, for example,

```
{Plus @@ AExpansion["terms"][[1 ;; 2]],
  Plus @@ Activate@AExpansion["termsInnerrθϕ"][[1 ;; 2]],
  Plus @@ Activate@AExpansion["termsInnerxyz"][[1 ;; 2]]} // ColumnForm
```

$$\frac{3\,V\,f[r,1]\,\text{LegendreP}[1,\text{Cos}[\theta]]}{2\,f[a,1]} - \frac{7\,V\,f[r,3]\,\text{LegendreP}[3,\text{Cos}[\theta]]}{8\,f[a,3]}$$

$$\frac{3\,r\,V\,\text{Cos}[\theta]}{2\,a} - \frac{7\,r^3\,V\,\left(-3\,\text{Cos}[\theta]+5\,\text{Cos}[\theta]^3\right)}{16\,a^3}$$

$$\frac{3\,V\,z}{2\,a} + \frac{7\,V\,\left(3\,x^2\,z+3\,y^2\,z-2\,z^3\right)}{16\,a^3}$$

A fairly large number of terms are required to achieve an accurate result near r = 1.

```
Block[{a = 1, V = 1, θ = 0, lab, Fz, G},
 (* Note use of Block rather than Module. This allows the numerical
  values of parameters a and V to be bound into the function Fz *)
 Fz[n_] := Plus @@ Activate@AExpansion["termsInnerxyz"]⟦1 ;; n⟧ /. {x → 0, y → 0};
 lab =
  Stl[StringForm["Potential inside sphere along +z-axis\nexact vs approximations"]];
 G[1] = Plot[{ Fz[10], Fz[11] , ΦSphere[z, a, V] }, {z, 0, a},
   AxesLabel → {Stl["z"], Stl["Potential"]}, PlotStyle → {Blue, Green, Black},
   PlotLegends → {"ΦIntsphere", "10 term approx", "11 term approx"},
   PlotLabel → lab, ImageSize → {400, 200}];

 Fz[n_] := Plus @@ Activate@AExpansion["termsOuterxyz"]⟦1 ;; n⟧ /. {x → 0, y → 0};
 lab = Stl[StringForm["Potential outside sphere along +z-axis"]];
 G[2] =
  Plot[{ Fz[10], Fz[11]}, {z, a, 3 a}, AxesLabel → {Stl["z"], Stl["Potential"]},
   PlotStyle → {Blue, Green}, PlotLegends → {"10 term approx", "11 term approx"},
   PlotLabel → lab, ImageSize → {400, 200}];

 Grid[{{G[1]}, {G[2]}}]]
```





I show a contour plot for the inner solution

```
Block[{a = 1, V = 1, contours, lab, Fxz, G},
 (* Note use of Block rather than Module. This allows the numerical
  values of parameters a and V to be bound into the function Fxz *)
 contours = Range[-0.9, 0.9, 0.2];
 Fxz[n_] := Plus @@ Activate@AExpansion["termsInnerxyz"][[1 ;; n]] /. y → 0;
 lab = Stl@StringForm["`` expansion terms", 3];
 G[1] = ContourPlot[Fxz[3], {x, -1, 1}, {z, -1, 1}, Contours → contours,
   RegionFunction → Function[{x, z}, x^2 + z^2 < 1], PlotLabel → lab];
 lab = Stl@StringForm["`` expansion terms", 7];
 G[2] = ContourPlot[Fxz[7], {x, -1, 1}, {z, -1, 1}, Contours → contours,
   RegionFunction → Function[{x, z}, x^2 + z^2 < 1], PlotLabel → lab];
 Grid[{{G[1], G[2]}}]]
```



and for the outer solution

```
Block[{a = 1, V = 1, contours, lab, Fxz, G},
 (* Note use of Block rather than Module. This allows the numerical
  values of parameters a and V to be bound into the function Fxz *)
 contours = Range[-0.9, 0.9, 0.2];
 Fxz[n_] := Plus @@ Activate@AExpansion["termsOuterxyz"][[1 ;; n]] /. y → 0;
 lab = Stl@StringForm["`` expansion terms", 3];
 G[1] = ContourPlot[Fxz[3], {x, -3, 3}, {z, -3, 3}, Contours → contours,
   RegionFunction → Function[{x, z}, x^2 + z^2 > 1], PlotLabel → lab];
 lab = Stl@StringForm["`` expansion terms", 7];
 G[2] = ContourPlot[Fxz[7], {x, -3, 3}, {z, -3, 3}, Contours → contours,
   RegionFunction → Function[{x, z}, x^2 + z^2 > 1], PlotLabel → lab];
 Grid[{{G[1], G[2]}}]]
```



The solution is quite good except near r = a

For completeness, I will generate electric field plots

```
Block[{a = 1, V = 1, n = 3, term, 𝔼xz, G},
 (* Note use of Block rather than Module. This allows the numerical
  values of parameters a and V to be bound into the function Fxz *)
 term = Plus @@ Activate@AExpansion["termsInnerxyz"][[1 ;; n]];
 𝔼xz = -Grad[term , {x, y, z}, "Cartesian"] /. y → 0;
 𝔼xz = 𝔼xz [[{1, 3}]];
 G[1] = VectorPlot[𝔼xz, {x, -1, 1}, {z, -1, 1}, RegionFunction →
    Function[{x, z, vx, vz, n}, x^2 + z^2 < 1], PlotRange → {{-3, 3}, {-3, 3}}];

 term = Plus @@ Activate@AExpansion["termsOuterxyz"][[1 ;; n]];
 𝔼xz = -Grad[term , {x, y, z}, "Cartesian"] /. y → 0;
 𝔼xz = 𝔼xz [[{1, 3}]];
 G[2] = Show[G[1],
   Graphics[{Blue, Circle[{0, 0}, a, {0, π}], Red, Circle[{0, 0}, a, {π, 2 π}]}],
   VectorPlot[If[x^2 + z^2 < a^2, 0, 𝔼xz], {x, -3, 3}, {z, -3, 3},
    RegionFunction → Function[{x, z, vx, vz, n}, x^2 + z^2 > 1]]]]
```



Appendix A: Visualizations

In[10]:=
```
Clear[ShowCoordinateSystemForGreensFunction];
ShowCoordinateSystemForGreensFunction[] :=
 Module[{a = 1, δ = 0.05, xMax = 1.5, O = {0, 0, 0}, axes, θarc, ϕarc, ξarc, 𝕒, 𝕣, 𝕣h,
   𝕣projection, OP, line𝕒, line𝕣, line𝕣h, lineBase, range3D , G, StoC, PointOnArc},
  StoC[r_, θ_, ϕ_] := {r Sin[θ] Cos[ϕ], r Sin[θ] Sin[ϕ], r Cos[θ] };
  (* point on the arc connecting points P1 and P2*)
  PointOnArc[α_(*0<α<1*), P1_, P2_, rMag_(*distance of arc from O*)] :=
   Module[{Pα = P1 + α (P2 - P1)}, rMag Pα / Norm[Pα]];

  range3D = {{-xMax, xMax}, {-xMax, xMax}, {-xMax, xMax}};
  (* spherical octant *)
  G[0] = ParametricPlot3D[{a {Cos[ϕ] Sin[θ], Sin[ϕ] Sin[θ], Cos[θ]}}, {ϕ, 0, 2 π},
    {θ, 0, π}, Mesh → False, ViewPoint → {2.4, 1.0, 1.25}, PlotStyle → Opacity[.25],
    Axes → None, Boxed → False, ImagePadding → 10, PlotRange → range3D];

  𝕒 = StoC[a, -π/4, π/4];
  𝕣 = StoC[2, π/4, π/4];
  𝕣projection = StoC[1, π/2, π/4];
  𝕣h = StoC[1.5, 3π/8, 3π/8] ;

  axes = {{Line[{{0, 0, 0}, {1.5, 0, 0}}], Text["x", 1.05 {1.5, 0, 0} ]},
    {Line[{{0, 0, 0}, {0, 1.5, 0}}], Text["y", 1.05 {0, 1.5, 0} ]},
    {Line[{{0, 0, 0}, {0, 0, 1.5}}], Text["z", 1.05 {0, 0, 1.5} ]}};

   line𝕒 = {Black, Arrowheads[Small], Arrow[{O, 𝕒}], Text[Stl["a"], 1.2 𝕒]};
  line𝕣 = {Black, Arrowheads[Small], Arrow[{O, 𝕒}], Arrow[{O, 𝕣}],
    PointSize[0.02], Point[𝕣], Text[Stl["𝕣   (r, θ, ϕ)"], 1.2 𝕣]};
  line𝕣h = {Black, Arrowheads[Small], Arrow[{O, 𝕣h}], PointSize[0.02],
    Point[𝕣h], Text[Stl["𝕣̂   (r̂, θ̂, ϕ̂)"], 1.2 𝕣h + {0, 0, 0.2}]};
  lineBase = {Dashed, Line[{𝕣, 𝕣projection}], Line[{O, 𝕣projection}]};
  θarc = With[{θVals = StoC[0.2, #, π / 4] & /@ Range[0, π / 4, π / 64]},
    {Black, Arrowheads[Small], Arrow[θVals], Text["θ", 1.6 θVals[[8]] ]}];
  ϕarc = With[{ϕVals = StoC[0.2, π/2, #] & /@ Range[0, π / 4, π / 64]},
    {Black, Arrowheads[Small], Arrow[ϕVals], Text["ϕ", 1.6 ϕVals[[8]]]}];
  ξarc = With[{ξvals = PointOnArc[#, 𝕣, 𝕣h, 0.75] & /@ Range[0, 1, 0.1]},
    {Black, Arrowheads[Small], Arrow[ξvals], Text["ξ", 1.2 ξvals[[5]]]}];

  G[1] = Graphics3D[{axes, {θarc, ϕarc, ξarc}, { line𝕒, line𝕣, lineBase, line𝕣h}},
    Boxed → False, PlotRange → range3D];
  Show[G[0], G[1], ImageSize → 400]]
```

In[12]:=
```
ShowCoordinateSystemForGreensFunction[]
```

Out[12]=



```
Module[{xMax = 1.1, lab, G},
 lab = Stl@StringForm["Sphere with top held at +V and bottom at -V"];
 G[1] = ParametricPlot3D[ {Sin[θ] Cos[ϕ], Sin[θ] Sin[ϕ], Cos[θ]},
   {θ, 0 , π/2}, {ϕ, -π, π}, Mesh → False, Boxed → False,
   Axes → None, PlotStyle → {Blue, Specularity[White, 40]},
   PlotRange → {{-xMax, xMax}, {-xMax, xMax}, {-xMax, xMax}}, PlotLabel → lab];
 G[2] = ParametricPlot3D[ {Sin[θ] Cos[ϕ], Sin[θ] Sin[ϕ], Cos[θ]},
   {θ, -π/2, -π}, {ϕ, -π, π}, Mesh → False, Boxed → False,
   Axes → None, PlotStyle → {Red, Specularity[White, 40]},
   PlotRange → {{-xMax, xMax}, {-xMax, xMax}, {-xMax, xMax}}];
 G[3] = Show[{G[1], G[2]}]]
```

Sphere with top held at +V and bottom at –V